NASA TECHNICAL NOTE

N71-16590

NASA TN D-6151

NASA TN D-6151

ON THE DYNAMICS OF SHORT PRESSURE PROBES: SOME DESIGN FACTORS AFFECTING FREQUENCY RESPONSE

by Ted W. Nyland, David R. Englund, and Robert C. Anderson

Lewis Research Center
Cleveland, Ohio 44135

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • FEBRUARY 1971

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA TN D-6151		

4. Title and Subtitle	5. Report Date
ON THE DYNAMICS OF SHORT PRESSURE PROBES: SOME DESIGN FACTORS AFFECTING FREQUENCY RESPONSE	February 1971
	6. Performing Organization Code
7. Author(s)	8. Performing Organization Report No.
Ted. W. Nyland, David R. Englund, and Robert C. Anderson	E-5752
	10. Work Unit No.
9. Performing Organization Name and Address	126-61
Lewis Research Center National Aeronautics and Space Administration Cleveland, Ohio 44135	11. Contract or Grant No.
	13. Type of Report and Period Covered
12. Sponsoring Agency Name and Address	Technical Note
National Aeronautics and Space Administration Washington, D.C. 20546	14. Sponsoring Agency Code

15. Supplementary Notes

16. Abstract

Measurements were made of the frequency response of 2.54-cm-long probes with inside diameters of 0.325 and 0.160 cm. The effects on frequency response of variations in probe geometry were also measured. These variations in geometry were designed to permit a simultaneous measurement of time-average pressure and to protect the pressure transducer from particles entrained in the airstream. Measured frequency response showed good correlation with predictions made using the equations of Iberall and of Bergh and Tijdeman. A computer program for calculating the frequency response of a multiple-tube volume system using the equations of Bergh and Tijdeman is presented in an appendix.

17. Key Words (Suggested by Author(s))		18. Distribution Statement
Pressure measurements	Amplitude ratio	Unclassified - unlimited
Pressure probe response	Phase measurement	
Frequency response	Response testing	

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified	26	$3.00

*For sale by the National Technical Information Service, Springfield, Virginia 22151

ON THE DYNAMICS OF SHORT PRESSURE PROBES: SOME DESIGN FACTORS AFFECTING FREQUENCY RESPONSE

by Ted W. Nyland, David R. Englund, and Robert C. Anderson

Lewis Research Center

SUMMARY

Frequency response measurements of 2.54-centimeter-long probes with 0.325- and 0.160-centimeter diameters are described. Comparison of the measured response and that predicted from existing analyses showed good correlation. The results indicate that the Bergh and Tijdeman equations accurately predict the response of probes which include extension tubes for measuring time-average pressure. It is shown that the use of some type of baffle for particle impact protection which involves an enlargement of the entrance flow area does not reduce the resonant frequency of the probe. A computer program for calculation of probe response based on the Bergh and Tijdeman equation is given in the appendix.

INTRODUCTION

Investigations of the dynamic stall characteristic of turbojet engines have placed great emphasis on the measurement of transient total pressure. Experimenters wish to determine how and at what rate stall propagates through an engine. To obtain this information, many transient total pressure probes are located in the inlet and in between stages of the compressor on the engine under test. In general, the requirements placed on these transient measuring systems include a frequency-response flat (amplitude ratio within ±5 percent) to at least 500-hertz, operation in airstream temperatures ranging from room temperature to approximately 800 K and reliable operation where particulate matter may be entrained in the airstream. In addition, it is of interest to measure the average total pressure at each probe location. The average pressure may range between a few to 200 newtons per square centimeter, and the transient pressure amplitudes may reach 30 percent of these values.

Measuring systems used in these investigations have consisted of probes with internally mounted miniature pressure transducers connected to short total head tubes (ref. 1). Such probes are generally water cooled so that the transducer is maintained at a moderate and relatively constant temperature. To protect the transducer from damage by the entrained particles in the airstream, the transducers have been mounted at right angles to the axis of the probe, or offsets or baffles have been included in the internal geometry of the probe. The measurement of average total pressure has been accomplished by connecting each probe through small-diameter tubing to high-accuracy low-frequency-response pressure instrumentation located external to the engine. This approach has proved to be preferable to relying on the average value of the output of the miniature (transient) pressure transducers mainly because the zero stability of these transducers is not compatible with the accuracy required for these data. The problem faced by the probe designer is how to incorporate these features of particle impact protection and average total pressure measurement into the probe design without degrading the transient response of the measuring system.

The purpose of this report is to present practical design information related to optimizing the transient response of such probes. The information contained herein is the result of an experimental program at the Lewis Research Center. The major objectives of this program were

(1) To measure the frequency response of simulated transient pressure probes

(2) To measure the effect on response of variations in probe geometry aimed at providing particle impact protection and average pressure measuring capability

(3) To correlate measured response with existing theoretical analyses on transient response of probes.

There are a number of theoretical analyses that can be used to predict the transient response of simple probe configurations. Estimates of the natural frequency can be obtained by using the simple quarter wavelength (organ pipe) and other tube-and-volume analyses found in many textbooks (e.g., see refs. 2 and 3). These analyses are based on linear second-order models of the system. More rigorous analyses have been done by Iberall (ref. 4) and more recently by Bergh and Tijdeman (ref. 5). The Iberall analysis is limited to a single tube and volume; Bergh and Tijdeman extended this analysis to a series connection of a number of tubes and volumes. These later analyses are not based on second-order system models and consequently do not yield a simple expression for a resonant frequency. However, they do permit the calculation of the amplitude ratio and phase shift as a function of excitation frequency from which the frequency for maximum amplitude ratio, that is, the resonant frequency, can be obtained. In either case the useful frequency range of the probe is that portion of the frequency spectrum for which the amplitude ratio is equal to one to within some acceptable tolerance (±5 percent for our purposes). In cases where the damping is negligible, which is generally true for the probes considered in this report, the frequency at the 5-percent tolerance limit is approximately equal to 20 percent of the resonant frequency.

The analyses of references 4 and 5 have been verified experimentally for frequencies up to about 2000 hertz (refs. 5 and 6). Recently, Goldschmied (ref. 7) has further verified the analysis of reference 4. The data contained herein show that references 3 and 4 accurately predict the response of the short tubes tested in this work.

PROBE CONFIGURATIONS

The probe designs used in this work were intended to simulate only the internal geometry of typical total pressure probes. The outside diameter, wall thickness, and flat end of the probes were chosen for ease of mounting in the test apparatus and do not represent good design practice for total pressure probes. The exterior and inlet geometries of probes for actual measurements in a flowing stream should be governed by such considerations as flow angle variations, pressure gradients, probe blockage, etc.

Straight Probe

The basic (straight) probe tested in this work consisted of a short straight tube of constant inside diameter terminated with a cavity formed by the transducer mounting (see fig. 1). As the volume ratio, the ratio of cavity volume to tube volume, is a parameter in the analytical equations, provision was made for varying the cavity volume. In this report, the straight probes will be designated as either A1 or A2, referring to the diameters 0.325 or 0.160 centimeter, respectively, followed by a number corresponding to the volume ratio. A miniature quartz piezoelectric pressure transducer was used in this probe.

In order to provide a means to measure the average total pressure, the probe was provided with taps drilled at right angles to the probe axis as shown in figure 1. The tap diameter was either 0.150 or 0.071 centimeter. These matched the inside diameters of tubing used when testing the effects of average pressure measuring systems on probe response. The lengths of the average pressure tubing tested were 15, 61, and 183 centimeters.

In one variation of the basic probe design, the transducer was mounted at right angles to the probe axis as shown in figure 2. The cavity volume was cylindrically shaped with its axis perpendicular to the tube axis. A miniature semiconductor strain gage pressure transducer, 0.317 centimeter in diameter, was used in this probe configuration.

Baffled Probe

The baffled probe tested in this work was designed to provide particle impact protection for the transducer. The baffle was built into the opening of the probe as shown in figure 3. The baffle consisted of an annular passageway connected through a set of holes to the inside tube. Four similar configurations were tested, each having different ratios of annular area to tube area and hole area to tube area as shown in the figure.

For baffle probes B1, B2, and B3, a set of 5 holes, 0.139 centimeter in diameter, were drilled between the annulus and probe tube. For probe B4, two sets of holes were used. The volume ratio based on inside tube length was 0.118. The pressure transducer was of the miniature quartz variety.

TEST APPARATUS AND PROCEDURE

A sinusoidal pressure generator developed at the Lewis Research Center was used to determine the frequency response of the probes. A drawing of the generator is shown in figure 4, and a detailed description of the generator is given in reference 8. The generator consists of a 2.54-centimeter-diameter closed tube driven by an annular shape air jet. The frequency of oscillation is varied by moving a tuning piston in the tube, effectively changing the tube length. Frequencies between 300 and 5000 hertz can be obtained with the generator. The probes are flush mounted to the inside surface of the resonator tube. A second pressure transducer is flush mounted directly opposite the probes and is used as a reference for both the amplitude and phase angle measurements. All probes were tested at atmospheric pressure and at room temperature with nominal pressure amplitudes of 0.5 newton per square centimeter (5 percent of atmospheric) peak to peak. The estimated error in the amplitude ratios, which are less than 15 decibels is 2 to 3 percent. As the amplitude ratios increase beyond 15 decibels, the error in the ratio increases to 10 to 20 percent at 30 decibels. Estimated error for the frequency measurement is ±1 hertz and for the phase angle measurement is $\pm 5^{\circ}$.

TEST RESULTS

Effects of Volume Ratio on Probe Response

The measured frequency response of probe A1-0.027 is shown in figure 5. Also shown is the analytical response curve for the probe as calculated from reference 4. (The values of fluid parameters used in all calculations are given in table I.) The analytical curve includes an end correction of $8D/3\pi$, as suggested in reference 3, added to the length term (D is tube diameter). Without the end correction, the analytical curve is the same shape, but it has a resonant frequency approximately 10 percent higher than the curve shown. Generally, the measured resonant frequencies of all the A1 and A2 probes were within 5 percent of the end corrected analytical values.

The effects of changes in volume ratio on the resonant frequency are shown in figure 6. Plotted is the percent reduction in resonant frequency as a function of volume ratio as calculated from reference 4 with end correction. The reduction in resonant frequency is calculated as the difference between the resonant frequency of a probe with (F_V) and without (F_O) a volume at the transducer end. Calculations indicate that the resonant frequency is slightly dependent on probe diameter. For the two probe diameters tested in this program, the resonant frequency for the same length and volume ratio differed by approximately 3 percent. For example, F_O for the A1 probe was 3000 hertz and for the A2 probe, 3100 hertz. As this is not a significant difference, the analytical curve plotted in figure 6 is the average value for the A1 and A2 probes. Also plotted in the figure are experimental data from a number of straight probe tests. For volume ratios less than 0.2, the experimental data are within 5 percent of the theoretical curves. Thus, it is concluded that for volume ratios less than 0.2, the response of a straight probe can be predicted to within 5 percent by the analysis of reference 4.

It is also of interest to determine how closely the quarter wavelength equation predicts the resonant frequency. The quarter wavelength equation with the $8D/3\pi$ length correction is

$$f = \frac{c}{4\left(L + \frac{8D}{3\pi}\right)}$$

where

f resonant frequency

c adiabatic velocity of sound

L length of tube

D inside diameter

For short probes with zero volume ratio, this equation predicts resonant frequencies which are higher than the reference 4 predictions by 2 to 5 percent. This agreement is adequate for most probe design work. If the probe has an appreciable volume ratio, a correction can be obtained from figure 6.

In this work, no attempt was made to compare the experimental results with predictions from other existing simplified tube-and-volume analyses. It is expected that many of these would also be adequate for probe design work, but would be of no advantage over the simple procedure outlined in the preceding paragraph.

From figure 6, it is apparent that a volume at the transducer end of the probe will decrease the resonant frequency of a straight probe. To minimize this decrease, a volume ratio of less than 0.1 should be considered as a reasonable goal for a probe design. This would result in a resonant frequency decrease of less than 10 percent from the ideal case. It is obvious that a shorter tube length could be used to increase the resonant frequency so long as the L/D ratio is greater than 1. However, the mounting of a transducer in the probe will almost always result in some volume at the end of the probe. Thus, as the probe length is decreased, the volume ratio will increase, thereby reducing, to some extent, the gain in frequency response achieved by a shorter probe.

The frequency response of the probe with the pressure transducer mounted at right angles to the probe centerline is shown in figure 7. The analytical response from reference 4 with an end correction is also shown. A comparison of figure 7 with the curves in figure 5 indicates that the basic response of the probe has not been altered by the right angle transducer mount.

The Effects of Average Pressure Tubing on Probe Response

The response of probes with two different sizes of average pressure tubing is shown in figure 8. The basic probe was the A1-0.027; the average pressure tube was 61 centimeters long and was closed at the end. One average pressure tube had a 0.071-centimeter inside diameter; the other had a 0.150-centimeter inside diameter. The response of the probe with the smaller tube is negligibly different in the usable frequency range from the probe without the average pressure tube (see fig. 5 for comparison). The only apparent difference is a slight reduction in the peak amplitude ratio. The effect of the larger diameter tube is more pronounced. The peak amplitude ratio is considerably lower and the slope of the phase angle curve is reduced in the region of the resonant frequency. These changes are similar to the effects of added damping in a linear second-order system. One other effect is present: the amplitude ratio plot for this probe has small but measurable cyclic variations as a function of frequency. These variations are caused by standing waves in the average pressure tube interacting with the probe.

Other tests were run with average pressure tubing in which the end of the tube was open to the atmosphere or terminated in a volume of 1000 cubic centimeters. The results were similar to the closed tube results with the exception that the phase of the variations in amplitude ratio was reversed. Where a maximum amplitude ratio occurred for a closed tube, a minimum amplitude ratio occurred for the open and volume terminated tubes.

The cyclic variation in amplitude ratio for a probe with average pressure tubing can be predicted from the Bergh and Tijdeman equations of reference 5. This is shown in figure 9 where the computed response, using the subroutines in the appendix, is compared with the experimental data for an A1-0.108 probe which includes an average pressure tube 61 centimeters long with a 0.150-centimeter inside diameter, closed at its end. Figure 9(a) shows the response through the first resonance, and figure 9(b) shows an expanded plot of the response below 1 kilohertz. For the analytical curve, it was assumed that the average pressure tubing was connected directly into the cavity volume. The agreement between the predicted response and the measured data is good. Examination of figure 9(a) will show that successive peaks or valleys in the amplitude ratio curve are separated by about 270 hertz, which is roughly equal to the half-wavelength frequency of a 61-centimeter tube in room-temperature air. This supports the conclusion that the variations in the amplitude ratio of the probe are caused by standing waves in the average pressure tube.

These cyclic variations in amplitude ratio represent a source of error in transient measurements. It is therefore of interest to determine how to limit the magnitude of these variations to within acceptable bounds. This might be done by reducing the diameter of the average pressure tube to the extent that viscous losses prevent the buildup of standing waves. The requirement for this is that $D < \sqrt{2\mu/\rho\pi f}$ (ref. 2, p. 229) where D is the inside diameter of the tube, μ is the dynamic viscosity, ρ is the density, and f is the frequency (in Hz). This approach leads to tube diameters of less than 0.025 centimeter, small enough to be prone to plugging. Such diameters are apparently smaller than necessary based on the results shown in figure 8. This is probably because this approach does not take into account that the pressure waves in the averaging tube must interact with the probe impedance to produce measurable variations. The probe impedance is relatively low because of large diameter of the probe.

An alternative approach is to determine a relation between the magnitude of the variations in amplitude ratio and the geometry of the probe and average pressure tube. This was done by calculating the response of a number of different geometries and empirically finding a geometric parameter that correlates with the magnitude of the cyclic deviations in amplitude ratio. The geometries covered in this calculation are shown in table II. The calculations were done only for air at atmospheric pressure and room temperature.

Figure 10 shows the results of these calculations. The average value of the peak-to-peak cyclic deviations in amplitude ratio over the usable frequency range (in this case up to 600 Hz) for each geometry is plotted against an empirical geometry parameter. The parameter found to give a reasonable correlation is $D_p(L_A + 13)/D_A^2$, where

D_P is the inside diameter of the probe, L_A is the length, and D_A is the inside diameter of the average pressure tube (all dimensions are in centimeters). The line drawn to the right of all the data points represents a conservative estimate of the geometry parameter necessary to insure that the cyclic deviations in the amplitude ratio are less than a given value. For example, to insure that these deviations are less than 2 percent (i.e., a ±1 percent error in amplitude ratio), a geometry parameter of at least 2000 is necessary. Thus, an A1-0 probe with an average pressure tube diameter of 0.150 centimeter would require a tube at least 125 centimeters long, but, if the tube diameter were 0.071 centimeter, the minimum length required would be 18 centimeters.

Also plotted in figure 10 are two measured data points for probes A1-0.027 and A1-0.108. The agreement between these measured data points and the calculations indicate that the results of figure 10 are applicable to probes with volume ratios up to at least 0.1.

Effects of Entrance Baffles on Probe Response

Test data for the baffled probe configurations are shown in figure 11, and the measured resonant frequencies are listed in figure 3. For comparison purposes the measured resonant frequency of an A1-0.108 probe is also listed in figure 3. (The volume ratio for the baffled probe configurations is approximately 0.13.) These data indicate that, if the entrance annulus and hole areas are larger than the probe tube area, this baffle configuration does not degrade the frequency response of the probe.

The slight increase in resonant frequency with increasing baffle area ratios is in qualitative agreement with the results of reference 5. (See figs. 19 and 20 of that ref.) The reference 5 data show that for a long tube-and-volume pressure transmission system, the resonant frequency can be nearly doubled by replacing half of the tubing length at the entrance end with a tube whose diameter is 1.5 times the diameter of the original tube. This would seem to imply that similar increases in resonant frequency might be obtainable for baffled probes of the type tested here by optimum choice of baffle area ratios and relative length. Preliminary tests of this concept indicate that moderate increases (10 to 20 percent) in resonant frequency are possible. However, practical limits on baffle size are such that a large increase in resonant frequency is not possible.

CONCLUDING REMARKS

The frequency response of short transient total-pressure probe geometries has been measured. Variations in geometry intended to provide particle impact protection and time-average total-pressure measuring capability have been studied to determine their effect on probe response. The measured response data have been compared with response predictions from various theoretical analyses. The results of this work are presented for the purpose of facilitating the design of practical transient total pressure probes for turbojet engine research.

The analyses of Iberall (ref. 4) and Bergh and Tijdeman (ref. 5) adequately calculate the amplitude ratio and phase angle versus frequency for the probes tested. The prediction of the resonant frequency is within 5 percent for probes with resonant frequencies up to 3 kilohertz.

A value of 0.1 for the ratio of transducer cavity volume to probe tube volume was felt to be a reasonable design goal; this volume ratio decreases the probe resonant frequency by less than 10 percent from the ideal case of zero transducer cavity volume.

For the probes tested, the simple, well-known quarter wavelength equation was adequate for calculating probe resonant frequency. If used with a length correction of $8D/3\pi$ where D is probe inside diameter, this equation could predict the resonant frequency of probes with zero volume ratio to within 10 percent. Further corrections for volume ratio can be made by information presented herein.

The method of using a small diameter tube between the transducer end of the probe and remote steady-state pressure instrumentation in order to measure time-average total pressure was tested. An empirical parameter involving the probe diameter and the average pressure tube diameter and length was found which adequately correlates with interaction effects between the average pressure tube and the probe. A design for which the value of this parameter is greater than 2000 will insure that the error due to this interaction is less than ±1 percent for a probe operated at atmospheric pressure and room temperature.

A baffle design for preventing particles entrained in the air stream from damaging the transient pressure transducer was tested. It was shown that with proper entrance area this baffle design does not degrade the frequency response of the probe.

Lewis Research Center,
National Aeronautics and Space Administration,
Cleveland, Ohio, September 28, 1970,
126-61.

APPENDIX - COMPUTER SUBROUTINE FOR CALCULATION OF THE FREQUENCY RESPONSE OF A MULTIPLE TUBE-VOLUME SYSTEM

For computing purposes, a system of tubes and volumes as in figure 12 was assumed. Using the Bergh and Tijdeman (ref. 5) results as a basis, the following equations were used for calculation:

$$\frac{P_j}{P_{j-1}} = \left[\cosh(\varphi_j L_j) + \frac{V_{v_j}}{V_{t_j}}(1 + \sigma_j) n_j \varphi_j L_j \sinh(\varphi_j L_j)\right.$$

$$\left. + \frac{D_{j+1}^2 \varphi_{j+1} J_2(\alpha_{j+1}) J_o(\alpha_j) \sinh(\varphi_j L_j)}{D_j^2 \varphi_j J_o(\alpha_{j+1}) J_2(\alpha_j) \sinh(\varphi_{j+1} L_{j+1})}\left(\cosh(\varphi_{j+1} L_{j+1}) - \frac{P_{j+1}}{P_j}\right)\right]^{-1}$$

where

$$\varphi_j = \frac{\omega}{\sqrt{\dfrac{\gamma P_s}{\rho_{s_j}}}} \sqrt{\frac{J_o(\alpha_j)}{J_2(\alpha_j)}} \sqrt{\frac{\gamma}{n_j}}$$

$$\alpha_j = i^{3/2} \sqrt{\frac{\omega D_j^2 \rho_{s_j}}{4\mu_j}}$$

$$n_j = \frac{1}{1 + \dfrac{\gamma - 1}{\gamma} \dfrac{J_2\left(\alpha_j \sqrt{Pr}\right)}{J_0\left(\alpha_j \sqrt{Pr}\right)}}$$

The variables are defined as follows:

Engineering symbol	Computer symbol	Definition
D	D	tube diameter
f	FREQ	frequency
J_n	--------	Bessel function of first kind of order n
L	XL	tube length
P	--------	amplitude of pressure distributions
Pr	P	Prandtl number (in subroutine usually equal to 0.7)
P_s	AMPRES	mean pressure
V_t	--------	tube volume
V_v	C	cavity volume
γ	--------	specific heat ratio
μ	VIS	fluid dynamic viscosity
ρ_s	RHO	mean mass density
σ	DVDP	dimensionless increase in cavity volume due to containment wall deflection
$\omega = 2\pi f$	--------	angular frequency
j	I	defines tube and volume

The FORTRAN IV subroutine using the preceding equations may be referenced as follows:

```
CALL AMPPHS(FREQ, A, PA, I, N)
```

FREQ is the pressure oscillation frequency for which the amplitude ratio and phase angle are to be calculated and must be specified before the CALL. The program places the amplitude ratio P(I)/P(0) in A and the phase angle in PA. N is the total number of series-connected tube-volume systems.

The system parameters must be put into the program through two common blocks referenced as follows:

```
COMMON/GEOMTY/D(10),XL(10),V(10),DVDP(10)

COMMON/THERMO/GAMMA, P(10),RHO(10),VIS(10),AMPRES
```

The program accepts as many as 10 series-connected systems with diameter (D), length (XL), volume (V), Prandtl number (P), and volumetric displacement (DVDP) specified in one block. The thermodynamic properties density (RHO), viscosity (VIS), ambient pressure (AMPRES), and specific heat ratio (GAMMA) are specified in a second block.

The program used for generating the Bessel functions is limited in use to Bessel functions of integer order with arguments of the form C(1 - i) where C is a real constant. The listing for the subroutines follow:

```
      SUBROUTINE AMPPHS(FREQ,A,PA,I,J)
C
C
C     THIS ROUTINE CALCULATES THE AMPLITUDE RATIO (A) P(I)/P(0)
C  AND PHASE ANGLE (PA)  FOR A SERIES CONNECTION OF J TUBE-VOLUME
C  SYSTEMS GIVEN FREQUENCY  'FREQ'.
C     THE SYSTEM GEOMETRY IS ENTERED THROUGH COMMON BLOCK /GEOMTY/.
C     THE THERMODYNAMIC CONDITIONS ARE ENTERED THROUGH COMMON BLOCK /THERMO/.
C
C
      DIMENSION PRATIO(10)
      COMMON/GEOMTY/D(10),XL(10),V(10),DVDP(10)
      COMMON/THERMO/GAMMA,P(10),RHO(10),VIS(10),AMPRES
      COMMON/XFREQ/FREQ1
      COMPLEX  PRATIO,RECURS
      FREQ1 = FREQ
      RECURS = 1.0
      DO 10  M=1,10
10     PRATIO(M) = 0.0
      N = J
C
C CALCULATE SUCCESSIVE PRESSURE RATIOS
C
      DO 20  K=1,N
20    CALL RATIO(PRATIO,J+1-K,J)
C
C CALCULATE REQUIRED RATIO BY MULTIPLYING APPROPRIATE PRESSURE RATIOS
C
      DO30 K = 1,I
30    RECURS = RECURS*PRATIO(K)
      A= CABS(RECURS)
      PA= ATAN2(AIMAG(RECURS),REAL(RECURS))
C
C CONVERT RADIANS TO DEGREES
C
      PA= PA*57.2957795
      IF(PA.GT.0.0) PA = -360.+PA
      RETURN
      END
```

```
$IBFTC XRATIO  DECK
       SUBROUTINE RATIO(PRATIO,J,K)
C
C
C    THIS PROGRAM CALCULATES THE PRESSURE RATIO P(J)/P(J-1).
C
C
       DIMENSION PRATIO(10),VRATIO(10),CORRCT(10)
       COMMON/GEOMTY/D(10),XL(10),V(10),DVDP(10)
       COMMON/THERMO/GAMMA,P(10),RHO(10),VIS(10),AMPRES
       COMMON/XFREQ/FREQ
       COMPLEX PHI,JRAT,POLY,POLRAT,CCOSH,CSINH,PRATIO,TERM1,TERM2,TERM3,
      CTPHI1,TPHI2,JRATIO
       PI = 3.1415927
       VRATIO(J) = V(J)/(PI*D(J)**2*XL(J)/4.)
       RFREQ = 2.*PI*FREQ
       IF(V(J).NE.0.0)GOTO1
       CORRCT(J) = 1.0
       GOTO3
1      CORRCT(J) = 1.0 + (DVDP(J)*AMPRES/V(J))
3      CONTINUE
       IF(J.EQ.K) GO TO 5
       POLRAT = CSQRT(POLY(J)/POLY(J+1))
       RHORAT = SQRT(RHO(J+1)/RHO(J))
       VOLRAT = D(J+1)**2/(D(J)**2)
       TPHI2 = PHI(J+1)
       JRAT = CSQRT(
      JJRATIO(CSQRT(CMPLX(0.0,-(RFREQ*D(J+1)**2*RHO(J+1)/
      J( 4.0*VIS(J+1)))))))/
      JJRATIO(CSQRT(CMPLX(0.0,-(RFREQ*D(J  )**2*RHO(J  )/
      J( 4.0*VIS(J  )))))))
5      TPHI1 = PHI(J)
       TERM1 = CCOSH(TPHI1*XL(J))
       TERM2 = VRATIO(J)*CORRCT(J)*XL(J)*TPHI1*POLY(J)*CSINH(TPHI1*XL(J))
       IF(J.EQ.K) GO TO 2
       TERM3 = (CSINH(TPHI1*XL(J))/CSINH(TPHI2*XL(J+1)))*
      T        VOLRAT*RHORAT*JRAT*POLRAT*
      T        (CCOSH(TPHI2*XL(J+1))-PRATIO(J+1))
       GOTO6
2      TERM3 = CMPLX(0.0,0.0)
6      CONTINUE
       PRATIO(J) = 1.0/(TERM1+TERM2+TERM3)
       RETURN
       END
```

```
$IBFTC XJRATO  DECK
       COMPLEX FUNCTION JRATIO(CARG)
C
C
C  THIS ROUTINE CALCULATES THE RATIO OF THE BESSEL FUNCTION OF THE
C FIRST KIND, ORDER 2 TO THE BESSEL FUNCTION OF THE FIRST KIND,
C ORDER 0 WITH COMPLEX ARGUMENT.
C
```

```
      COMPLEX FM,QSUM,PSUM,XPSUM(2),XQSUM(2),XJ(2),RATIO,XX,Y
     1Y,Z
      COMPLEX J2,J0,JRATIO,CARG
      COMMON /S1/X,Y,PSUM,QSUM
      COMMON /FMNAM/FM
      DATA PI/3.1415927/,M/-1/,Z/(0.0,1.0)/
      IF(REAL(CARG).GT.30.0) GO TO 40
C
C    IF THE REAL PART OF THE ARGUMENT IS GREATER THAN 30 USE IS MADE OF
C HANKELS ASYMPTOTIC EXPANSIONS (SEE NBS HANDBOOK OF MATHEMATICAL
C FUNCTIONS AMS NO. 55) TO CALCULATE THE BESSEL FUNCTION RATIO. THE P
C AND Q COEFFICIENTS USED IN THE EXPANSIONS ARE CALCULATED USING THE
C SERIES SUBROUTINE.
C
      GO TO 20
40    X = REAL(CARG)
      Y = -X
      DO 2 J=1,2
      M = M + J
      XI = FLOAT(M)
      FM = CMPLX(XI,0.0)
      CALL SERIES
      XPSUM(J) = PSUM
2     XQSUM(J) = QSUM
      DO 3 K=1,2
      A = X + (0.75 - FLOAT(K))*PI
      XX = XPSUM(K) * COS(A)  - XQSUM(K) * SIN(A)
      YY = (XPSUM(K) * SIN(A) + XQSUM(K) * COS(A)) * Z
3     XJ(K) = XX + YY
       RATIO = XJ(2) / XJ(1)
      JRATIO = RATIO
      M = -1
      GO TO 30
20    CALL ZEEBES(J0,0,CARG)
      CALL ZEEBES(J2,2,CARG)
      JRATIO = J2/J0
30    CONTINUE
      RETURN
      END
```

```
$IBFTC SERIE   DECK

      SUBROUTINE SERIES
C
C
C     THIS ROUTINE CALCULATES THE VALUE OF THE SERIES NEEDED FOR
C HANKELS ASYMPTOTIC EXPANSION
C
C   IN CALCULATING P(J) AND Q(J) NUMBERS WITH EXPONENTS LESS THAN -38
C MAY OCCUR.  THIS PROGRAM USES A BUILT-IN ROUTINE WHICH SETS P(J)
C AND/OR Q(J) EQUAL TO ZERO WHENEVER THESE SMALL NUMBERS OCCUR.
C
C
C
```

```
      COMPLEX Z,MU,PSUM,QSUM,FM,P(50),Q(50)
      COMMON/S1/X,Y,PSUM,QSUM
      COMMON /FMNAM/FM
      PSUM = (0.,0.)
      QSUM = (0.,0.)
      MU =4.*FM**2
      Z = CMPLX(X,Y)
      DO 100 J=1,50
      N = J-1
      FN =N
      IF(N .NE. 0) GO TO 30
20    P(J) =(1.,0.)
      Q(J) = (MU-1.)/(8.*Z)
      GO TO 90
30    FN2 = 2.*FN
      FN4 = 2.*FN2
      FN21 = FN2 + 1.
      FN41 = 2.*FN21
      IF(CABS(P(J-1)).EQ.0.) GO TO 50
      P(J) = -((MU-(FN4-3.)**2)/(64.*(FN2-1.)*FN2))*((MU-(FN4-1.)**2)/Z
     $**2)*P(J-1)
      IF (J .GT. 2 .AND. CABS(P(J)) .GT. CABS(P(J-1)) ) P(J) = (0.,0.)
      GO TO 55
50    P(J) = (0.,0.)
55    IF(CABS(Q(J-1)).EQ.0.) GO TO 80
      Q(J) = -((MU-(FN41-3.)**2)/(64.*FN2*FN21))*((MU-(FN41-1.)**2)/Z
     $**2)*Q(J-1)
      IF (J .GT. 2 .AND. CABS(Q(J)) .GT. CABS(Q(J-1)) ) Q(J) = (0.,0.)
      GO TO 90
80    Q(J) = (0.,0.)
90    PSUM = P(J)+PSUM
      QSUM = Q(J)+QSUM
      IF(CABS(P(J)).EQ.0. .AND. CABS(Q(J))  .EQ. 0.) GO TO 110
100   CONTINUE
110   CONTINUE
      RETURN
      END
```

```
$IBFTC BESSNN  DECK

      SUBROUTINE ZEEBES(J,N,Z)

C
C   THIS ROUTINE CALCULATES THE VALUE OF THE BESSEL FUNCTION OF THE
C FIRST KIND OF INTEGER ORDER. THE ARGUMENT IS OF THE FORM
C C(1 - SQRT(-1)) WHERE C IS A REAL CONSTANT LESS THAN 30 .
C
C                    J...BESSEL FUNCTION VALUE
C                    N....BESSEL FUNCTION ORDER
C                    Z...BESSEL FUNCTION ARGUMENT
C
C
      COMPLEX J,Z
      DOUBLE PRECISION DFN,T,R(2),C,DK,D,B
      D = REAL(Z)
```

```
      C = D * 1.4142135623730950D0
      CR = D
      DFN = FLOAT(N)
      IF(N.NE.2) GO TO 1
      T = 5.D-1
      GO TO 2
1     T = 1.D0
2     R(1) = T
      R(2) = 0.D0
      C = (C**2.) / 4.D0
      DK = 0.D0
      I = 1
      M = 0
5     DK = DK + 1.D0
      T = T * C / ((DK)*(DK+DFN))
      I = I + 1
      IF (T.LT.DABS(R(I)*1.D-8)) GO TO 7
      M = M + 1
      B = 1.D0
      IF(M.GT.1) B = -1.D0
      IF(M.EQ.3) M = -1
      R(I) = R(I) + B * T
      IF( I.EQ.2) I = 0
      GO TO 5
7     RR= R(1)
      RI = R(2)
      IF(N.NE.2) GO TO 8
      RI = (-1.) * R(1) * C
      RR =R(2) * C
8     J = CMPLX(RR,RI)
      RETURN
      END


$IBFTC SUBPHI  DECK

      COMPLEX FUNCTION PHI(J)
C
C        THIS PROGRAM CALCULATES A FUNCTION PHI WHICH IS AN
C     ATTENUATION PARAMETER.
C
      COMMON/GEOMTY/D(10),XL(10),V(10),DVDP(10)
      COMMON/THERMO/GAMMA,P(10),RHO(10),VIS(10),AMPRES
      COMMON/XFREQ/FREQ
      COMPLEX POLY,PHI,JRATIO
      SN =6.2831854*FREQ*D(J)**2*RHO(J)/VIS(J)
      PHI=6.2831854*FREQ*SQRT(RHO(J)/(GAMMA*AMPRES))*
     P       CSQRT(1./JRATIO(CSQRT(CMPLX(0.0,-(SN/4.0)))))*
     P       CSQRT(GAMMA/POLY(J))
      RETURN
      END
```

```
$IBFTC SUBPLY  DECK

      COMPLEX FUNCTION POLY(J)
C
C        THIS ROUTINE CALCULATES A POLYTROPIC COEFFICIENT
C     (LABELED 'N' IN BERGH AND TIDJEMAN)
C
      COMMON/GEOMTY/D(10),XL(10),V(10),DVDP(10)
      COMMON/THERMO/GAMMA,P(10),RHO(10),VIS(10),AMPRES
      COMMON/XFREQ/FREQ
      COMPLEX JRATIO,CARG,POLY
      SN = 6.2831854*FREQ*D(J)**2*RHO(J)/VIS(J)
      ARG = P(J)*SN/4.0
      CARG= CSQRT(CMPLX(0.0,-ARG))
      POLY=1.0/(1.0+((GAMMA-1.0)/GAMMA)*JRATIO(CARG))
      RETURN
      END


$IBFTC FCCOSH  DECK

      COMPLEX FUNCTION CCOSH(ARG)
      COMPLEX ARG,CCOSH
      CCOSH = (CEXP(ARG)+CEXP(-ARG))/2.0
      RETURN
      END


$IBFTC FCSINH  DECK

      COMPLEX FUNCTION CSINH(ARG)
      COMPLEX ARG,CSINH
      CSINH= (CEXP(ARG)-CEXP(-ARG))/2.0
      RETURN
      END
```

REFERENCES

1. Armentrout, Everett C.: Development of a High-Frequency-Response Pressure-Sensing Rake for Turbofan Engine Tests. NASA TM X-1959, 1970.

2. Crandall, Irving B.: Theory of Vibrating Systems and Sound. D. Van Nostrand Co., Inc., 1926.

3. Doebelin, Ernest O.: Measurement Systems: Application and Design. McGraw-Hill Book Co., Inc., 1966.

4. Iberall, Arthur S.: Attenuation of Oscillatory Pressures in Instrument Lines. Nat. Bureau Standards J. Res., vol. 45, no. 1, July 1950, pp. 85-108.

5. Bergh, H.; and Tijdeman, H.: Theoretical and Experimental Results for the Dynamic Response of Pressure Measuring Systems. Rep. NLR-TR-F.238, National Aero- and Astronautical Research Inst., Amsterdam, Jan. 1965.

6. Watts, Geoffrey P.: An Experimental Verification of a Computer Program for the Calculation of Oscillatory Pressure Attenuation in Pneumatic Transmission Lines. Rep. LA-3199-MS, Los Alamos Scientific Lab., Feb. 10, 1965.

7. Goldschmied, F. R.: On the Frequency Response of Viscous Compressible Fluids as a Function of the Stokes Number. J. Basic Eng., vol. 92, no. 2, June 1970, pp. 333-347.

8. Nyland, Ted W.; Englund, David R., Jr.; and Gebben, Vernon D.: System for Testing Pressure Probes Using a Simple Sinusoidal Pressure Generator. NASA TM X-1981, 1970.

9. Hilsenrath, Joseph; Beckett, Charles W.; Benedict, William S.; Fano, Lilla; Hoge, Harold J.; Masi, Joseph F.; Nuttall, Ralph L.; Touloukian, Yerman S.; and Woolley, Harold W.: Tables of Thermal Properties of Gases. Circ. 564, National Bureau of Standards, Nov. 1, 1955.

TABLE I. - PROPERTIES OF AIR

USED IN COMPUTATIONS (REF. 9)

Temperature, K	295
Pressure, N	9.92
Density, g/cm^3	1.172×10^{-3}
Viscosity, g/(sec)(cm)	1.824×10^{-4}
Specific heat ratio	1.4

TABLE II. - PROBE GEOMETRIES USED IN

CALCULATING EFFECTS OF AVERAGE

PRESSURE TUBES ON PROBE RESPONSE

[Probe length, 2.54 cm; probe volume ratio, 0; average pressure tubing lengths, 25, 102, and 203 cm.]

Probe diameter, cm	Average pressure tube diameter, cm		
0.157	0.051	0.076	0.101
.317	.076	.127	.177
.476	.101	.152	.203



Figure 1. - Basic (straight) probe. (Dimensions are in centimeters.)



Figure 2. - Perpendicularly mounted transducer probe. (Dimensions are in centimeters.)



Probe	Dimensions		Area ratios		Measured resonant frequency, Hz
			Annulus area to tube area ratio	Hole area to tube area ratio	
B1	0.235	-----	0.87	0.92	2620
B2	.208	-----	1.30	.92	2770
B3	.177	-----	1.80	.92	2860
B4	.177	0.381	1.80	1.80	3030
AI-0.10	----	-----	----	----	2790

Figure 3. - Baffle probe design data. (Dimensions are in centimeters.)



Figure 4. - Pressure generator.



Figure 5. - Frequency response of probe A1-0.027.



Figure 6. - Percent reduction in resonant frequency as function of volume ratio. Analytical curve is the average values computed for A1 and A2 type of probes.



Figure 7. - Frequency response of perpendicularly mounted transducer probe.



Figure 8. - Frequency response of two A1-0.027 probes which include 61 centimeters of average pressure tubing.



(a) Response from 350 to 4500 hertz.

(b) Response from 350 to 1000 hertz.

Figure 9. - Frequency response of an A1-0.108 probe which includes 61 centimeters of 0.150-centimeter inside diameter tubing for average pressure measurements.



Figure 10. - Average cyclic deviation of amplitude ratio as function of geometry parameter for air at atmospheric pressure and room temperature.



Figure 11. - Frequency response of baffled probe configuration.



Figure 12. - Tube-volume system for the Bergh and Tijdeman equation.